Danmarks Statistik
MODELGRUPPEN

Arbejdspapir*

Morten Werner
Lars B. Termansen

13. juni 2001

# Dokumentation af LADA

**Resumé:**

*Papiret dokumenterer LADA modellen som den er endt med at se ud. Bemærk, at papiret er identisk med kapitel 3 i "Environmental satellite models for ADAM", hvilket forklarer den lidt specielle afsnits-, lignings- og tabelnummerering.*

MOW13601.doc

Nøgleord: landbrug, disaggregering af ADAM a-erhverv, Satellitmodeller

*Modelgruppepapirer er interne arbejdspapirer. De konklusioner, der drages i papirerne, er ikke endelige og kan være ændret inden opstillingen af nye modelversioner. Det henstilles derfor, at der kun citeres fra modelgruppepapirerne efter aftale med Danmarks Statistik.*

## 3. The LADA-model

The LADA model describes the production in 5 agricultural subsectors constituting the agricultural sector in ADAM. These subsectors are *crops*, *cattle and milk production*, *pigs, poultry* and a sector defined residually, named the *q*-sector. The LADA model has two main purposes. First, the model can be used as a translation and aggregation module, which translates ESMERALDA scenarios into LADA scenarios and aggregates these scenarios into a scenario describing the agricultural sector in ADAM. Secondly, the model can be used to analyse simple and small changes in the agricultural subsectors of the model compared to some baseline scenario. For this purpose LADA has a simple description of factor demand and land use in the subsectors. In both cases the LADA-model provides a complete scenario describing the agricultural sector in ADAM as well as projections of the physical production in ESMERALDAs 16 lines of production, which are used as input in the environmental satellite model describing emissions from agriculture.

Section 3.1 describes the data construction methods and sources, which have been used during the data construction. In section 3.2 the transformation of ESMERALDA scenario into LADA scenarios is described while section 3.3 contains a description of the modelling of the subsectors production and factor demand. Section 3.4 comments on different ways of using the model.

### *3.1 Data construction*

The five LADA subsectors: *crops*, *cattle and milk production*, *pigs*, *poultry* and the *q*-sector are a disaggregation of the ADAM agricultural sector, the *a*-sector. The objective in the data construction is to obtain subsector series for *crops*, *cattle and milk production*, *pigs* and *poultry*, which are consistent with an appropriate aggregation of the ESMERALDA lines of production. The consistency of the series is crucial when ESMERALDA scenarios are used for projection of the LADA subsectors.

The historical data concerning production and output prices in the five subsectors are published in The Agricultural Statistics and National Accounts from Statistics Denmark. Data on production and output prices describing 29 subsectors can be found in the Agricultural Statistics. These series are aggregated to the LADA subsectors crops, cattle pigs and poultry. Data from the National Accounts are mainly used to construct data describing the *q*-subsector. Table 3.1.1 below shows the connections of the LADA subsectors, the Agricultural Statistic, The National Accounts and the ESMERALDA lines of production. In Table 3.1.1 note the residual component of the *q*-subsector. This residual contains the difference between the definition of the agricultural sector in the Agricultural Statistics and the National Account compared to the agricultural sector in ADAM. Accordingly the *q*-subsector contains for instance fishing and forestry besides what is shown in Table 3.1.1.

Data concerning the input side of production i.e. use of energy and materials, labour and capital are constructed based on a disaggregation of the agricultural sector from the input-output tables published by Statistics Denmark into the five subsectors. The disaggregation has been performed for one year by the SJFI. Data for the remaining years have been constructed based on information of production and total input in the *a*-sector assuming that the production structure is fixed. Other sources in the data construction are historical data from SJFI and historical data from the ADAM model. Although the input side data series cannot be claimed to be historical they constitute a reasonable basis for projections of input use in the five subsectors based on ESMERALDA scenarios. A detailed description of the data and data constrution can be seen in Nielsen (2000) and Werner (2000a).

Table 3.1.1 Construction of production and output prices

| **LADA subsector** | **Agricultural Statistics and National Account** | **ESMERALDA, lines of production** |
|---|---|---|
| Crops<br>*av*-subsector | Total cereals, pulses ripened, seeds for sowing, seeds for manufacturing, sugar beets, potatoes | spring barley etc., winther barley, wheat, pulses, rape, seeds for sowing, potatoes, sugar beets, fallow |
| Cattle<br>*ak*-subsector | Milk, cattle, grass and green fodder, other crop products | dairy cattle, nurse cows, rearing cattle, calves, fodder beets, grass rotation |
| Pigs<br>*as*-subsector | Pigs | sows, baconers |
| Poultry<br>*ao*-subsector | Eggs for human consumption, poultry | Poultry |
| Others<br>*aq*-subsector | Vegetables, mushrooms, fruit and berries, flowers, potted plants, nusery products, horses, sheep, furred animals, game, other livestock, residual | |

### *3.2 Projections based on ESMERALDA scenarios*

One of the main purposes of the LADA model is to be able to translate forecasts and policy scenarios from the ESMERALDA model into scenarios of the agricultural sector in ADAM. Thereby enabling forecasts from ADAM to be based on SJFI scenarios for the agricultural sector and derivation of macroeconomic effects of agricultural and environmental policies affecting the agricultural subsectors. The ESMERALDA scenarios involve projections in both fixed and current prices. The linkage between the ESMERALDA and LADA series are modelled in a submodel of the LADA model. In the following this submodel will be referred to as *the transformation module*. The purpose of the module is to transform an ESMERALDA scenario into a projection of production and factor demand in the five LADA subsectors using as much information from the ESMERALDA scenario as possible. The transformation module is described in detail in Werner (2000b).

The methods of projecting the variables in the LADA subsectors based on ESMERALDA scenarios differ for different categories of variables. The *aq*- subsector has to be handled separately as there is no information about this subsector in the ESMERALDA scenario.

The main categories are:

1) Production
2) Input of energy and material and gross value added
3) Labour force
4) Capital input and investment
5) Taxes etc.
6) q-subsector
7) Physical units

The projection of *production* series is straightforward as aggregation of ESMERALDA production series across lines of production and types of output cause no problems. The LADA subsector productions are obtained using the last historical observation in a given production series from LADA as a base and then projecting this series using the growth rate in the corresponding series from the aggregated ESMERALDA scenario. This is done in fixed and current prices and the output prices are derived.

Projecting the use of *energy and material* in the LADA subsectors based on the ESMERALDA scenario is a bit more difficult. The reason is that the ESMERALDA input structure is based on the costs of different material and energy inputs while the LADA input structure - like the input structure in ADAM - is based on an input-output model. The inputs are then aggregated to energy or material costs.

Table 3.1.2 shows the ESMERALDA cost structure and how it is linked to the LADA input structure. The first column shows the ESMERALDA costs. The second column shows the components at the input-output level in LADA, which are affected by the different ESMERALDA cost components. The third column shows the cost component, which is finally affected by the ESMERALDA costs.

In the transformation module the projection of the LADA energy and material demand is carried out at the level found in the ADAM input-output system involving supplies from 19 industries and 15 import groups. This implies that the composition of the aggregated material use changes over time in each of the agricultural subsectors in LADA. Thereby macroeconomic effects from environmental policies aimed at certain inputs in agricultural production for example fertilizers or pesticides are easier to derive. Again growth rates from an aggregation of the ESMERALDA forecast are used for projecting LADA series. Inputs at the disaggregated LADA level, which are not affected by any ESMERALDA cost component are projected using the observed value in the last historical year.

*Gross Value Added* is determined residually from production, energy and material use and some tax variables commented on below.

Table 3.1.2 Linking ESMERALDA costs to LADA costs

| ESMERALDA | Input from sector | LADA cost |
|---|---|---|
| Seeds | *av, M0* | *Material* |
| Fertilizer / manure | *ak, as, nk, M2, M5* | - |
| Concentrated feeds | *av, nf , M0* | - |
| Fodder roots | *ak* | - |
| Pesticides | *nk, M5* | - |
| Energy | *ng, ne, M3k, M3q* | *Energy* |
| Other services | *qt, qq* | *Material* |
| Contract operations | *aq* | - |
| Green fodder | *ak* | - |
| Labour | | *Labour costs* |
| Insurance | *qq* | *Material* |
| Other costs | *aq, qq* | - |
| Maintenance, equipment | *nm* | - |
| Costs equipment | | *Capital costs, equipment* |
| Maintenance, buildings | *b* | *Material* |
| Costs building | | *Capital costs, buildings* |
| Maintenance, land | *qq* | *Material* |
| Abbreviations: *av* – crop subsector, *ak* – cattle and milk subsector, *as* – pig subsector, *aq* – other agricultural, *ng* – petroleum refineries, *ne* – public energy supply, *nf* – manufacturing of food, *nm* – manufacturing of macinery, *nt* – Shipyards etc., *nk* – manufacturing of chemicals, *b* – construction, *qh* – trade, *qt* - miscellaneous transport, *qq* – miscellaneous services, *M0* – imports from SITC group 0, *M3k* – imports of coal, *M3q* – Imports from SITC 3 other than coal and crude oil, *M5* imports from SITC 5 | | |

The ESMERALDA model projects the use of *labour* measured in hours worked as well as labour cost in each line of production. These series are used for projecting the corresponding LADA series using growth rates from the aggregated ESMERALDA scenario. From these projections further series concerning the use of labour in LADA such as hourly compensation and persons employed are derived. The partition of total employment into self-employed and wage earners is derived using ADAM assumptions on the development in hours worked per year and the share of self-employed to total employment.

The projections of series concerning *capital stocks* of buildings and machinery are the least reliable due to two particular circumstances. First, it is rather difficult to split the aggregated capital stock of equipment and buildings in the agricultural sector in ADAM into the corresponding series concerning the five LADA subsectors. Consequently the levels of capital stocks in the subsectors might not be appropriate. Secondly, the corresponding series for the use of equipment and buildings in ESMERALDA are difficult to link to the stock series in LADA. Nevertheless, these series are important as they describe the assumptions on technological development underlying the ESMERALDA scenarios.

Despite the problems, an attempt is made to derive projections of the LADA capital stock series from the ESMERALDA scenario. Again the level of capital stocks in the last historical year in the LADA data is projected using growth rates from ESMERALDA series on the total cost of using equipment and buildings respectively. Knowing the capital stocks, gross investment is determined using a relation describing the accumulation of capital and ADAM assumptions on capital depreciation. Investment prices are derived from the ESMERALDA forecast. Based on these investment prices and assumptions on interest rates etc., the user-costs of capital are determined.

One way of avoiding the problems involved in linking the capital series could be to ignore the use of capital in the LADA model, and use the ADAM factor demand equations to determine the development of the stock of equipment and buildings in the agricultural sector contingent on the production determined in ESMERALDA. However, crucial information on the assumptions on technological development underlying the ESMERALDA scenario might be lost unless this information is extracted from the ESMERALDA scenario and introduced in the ADAM equations in some other way.

ESMERALDA provide forecasts of the subsidies received by the subsectors. The LADA model describes value added *taxes*, custom, taxes and subsidies on products and other taxes and subsidies. When projecting the subsidies the growth rate of the subsidies is used. The taxes are by and large projected using assumptions from the ADAM model.

The ESMERALDA scenario contains no information on what is happening in the *q*-sector, which is the residual between the agricultural sector in ADAM and agriculture as defined by SJFI. Historically, the production in the *aq*-sector constitutes approximately 40%. of total volume produced in the agricultural sector in ADAM. The transformation module per default projects the production in this sector to keep its relative importance unchanged compared to the last historical year. However, this procedure will not always be appropriate. For instance, this practice will exaggerate the effects of the policy when studying effects of an agricultural policy aimed at reducing pig production. The explanation being that the growth rate in the *aq*-production will be affected by the changed pig production. In such an analysis one solution could be to project the activity in the *aq*-subsector using the changes in the remaining subsectors.

Finally, series describing the production in 14 of the ESMERALDA subsectors in *physical units* are copied unaltered to the LADA scenario. These series that are used in the emission models, are measured in tons of production in the crop subsectors and number of animals in the animal subsectors.

### *3.3 The LADA subsectors*

Besides the transformation module, the LADA model contains a description of production and factor demand in the five subsectors. This feature can be used when one wants to study environmental and macroeconomic effects following simple and small changes in the agricultural production at the subsector level compared to some baseline scenario. Below, this part of the model is described, however only main features and key equations of the model are explicitly commented. The entire model is found in Annexes 3.1 and 3.2.

It is assumed that technology in the five subsectors can be described by a Leontief production function but the determination of the production level differs among the subsectors. In the *as*-, *ao*- and *aq*-subsectors production is considered exogenous while production in the *av*- and *ak*-subsectors are determined by the land available to the sectors.

In the *as*-, *aq*- and *ao*-subsectors it is assumed that equipment, buildings, labour, material and energy are used as inputs in production. Combined with the technology assumption the production in these subsectors can be written:

$$(3.3.1)\quad fX<k> = \min\left(\frac{fKm<k>}{bkm<k>}, \frac{fKb<k>}{bkb<k>}, \frac{Hq<k>}{bhq<k>}, \frac{fVm<k>}{bvm<k>}, \frac{fVe<k>}{bve<k>}\right)$$

where $k=as, aq, ao$ and $fX<k>$ is production in subsector $k$ measured in fixed prices. $fKm<k>$ is use of machinery, $fKb<k>$ is use of building, $Hq<k>$ is labour input measured in hours, $fVm<k>$ is input of materials and $fVe<k>$ is energy inputs. Both energy and material inputs are measured in fixed prices. $bkm<k>$, $bkb<k>$, $hq<k>$, $bvm<k>$ and $bve<k>$ are tecnological coefficients of equipment, buildings, labour, material and energy respectively.

Keeping in mind that production is considered exogenous, equation (3.3.1) and an assumption of cost minimization yield the factor demands. Note especially, that the input coefficients concerning input of material and energy are not explicit variables in the model, but are determined as sums of input-output coefficients from the disaggregated level, as:

$$(3.3.2)\quad bve<k> = ang<k> + ane<k> + am3k<k> + am3q<k>$$

and

$$(3.3.3)\quad \begin{aligned} &bvm<k> = \sum_j a<j><k> \\ &j = av, ak, as, ao, aq, nf, nm, nt, nk, b, qh, qt, qq, m0, m2, m3k, m3q, m5, si \end{aligned}$$

where the $a<j><k>$ are input-output coefficients at the disaggregated level of inputs. As an example the coefficient *anmas* shows how much of the input to the pig-subsector, which originates from the *nm*-industry, *manufacturing of machinery*.

Determination of production in the *av*- and *ak*-subsectors is different, since it is assumed that land is used as an input in production and that the land available to the subsectors is the limiting factor in production.
Again the technology assumption implies that production is given by:
(3.3.4)

$$fX<h> = \min\left(\frac{nv<h>}{bnv<h>}, \frac{fKm<h>}{bkm<h>}, \frac{fKb<h>}{bkb<h>}, \frac{Hq<h>}{bhq<h>}, \frac{fVm<h>}{bvm<h>}, \frac{fVe<h>}{bve<h>}\right)$$

where $h=av, ak$ and $nv<h>$ is the land available to subsector $h$ measured in hectares and $bnv<h>$ is the technological coefficients associated with land in subsector $h$. The remaining notation is as above.

Since the amount of land available to each subsector is the limiting factor the production in subsector *h* is determined as

(3.3.5) $$fX<h> = \frac{nv<h>}{bnv<h>}$$

Demands for the remaining factors are determined from equation (3.3.4), (3.3.5) and the assumption of cost minimization. Equations (3.3.2) and (3.3.3) also applies to material and energy coefficents, $bvm<h>$ and $bve<h>$, in the *av*- and *ak*-subsectors.

The total amount of land available, *nv*, is considered exogenous. In the land allocation between the *av*- and *ak*-subsector it is assumed, that land lying fallow, *nvbr,* is exogenous and the use of land in the *ak*-subsector, *nvak*, is given by

(3.3.6) $$nvak = nvsh + nvvg + nvrf$$

where *nvsh*, *nvvg* and *nvrf* are land used for rotation grass, permanent grass and fodder roots respectively.

The land available to the *av*-subsector, *nvav*, is determined residually as

(3.3.7) $$nvav = nv - (nvak + nvbr)$$

This modelling of production and land used by the subsectors *av* and *ak* implies that increasing land use and thereby production in one sector leads to a decline in land use and thereby production in the other subsector given the total amount of land available and the amount of land lying fallow. This property of the LADA model is mimicing a corresponding property in the ESMERALDA model. Of course one can also change production by changing the total amount of land available or the area lying fallow. In these cases, however, one have to keep in mind that changing the area laid fallow will affect the subsidies received by the sector and a change in the total area available to agricultural production will influence the economy through various channels.

Given the land available to the *av*- and *ak*-subsectors the production volumes are determined from (3.3.5) and assuming cost minimization the demand for capital, labour, material and energy is easily derived from equation (3.3.4).

From equation (3.3.1) and (3.3.4) it is noticed that all production factors, except for land in the *av*- and *ak*-subsector are modelled as fully flexible. This is not an appropriate description of the demand for equipment and buildings, and implies that only small changes in production can be appropriately analysed directly in the LADA model. If the changes are sufficiently small it can be argued at least regarding equipment that most of the desired change in the capital stock can be gained by instantly changing investments. If one wants to analyse larger changes in production it is recommended that demand for capital are either determined by the factor demand equations in ADAM or that explicit assumptions concerning the reduction or growth of the capital stock are made.

The necessary investments consistent with the capital stocks are determined by the accumulation identity:

(3.3.8) $fI<q><k>= fK<q><k> -(1-bfi<q>va)fK<q><k>_{-1}$

where *q=m,b* denotes equipment and buildings respectively and *bfi<q>va* are the depreciation rate for capital of type *q* obtained from the relevant ADAM scenario. Finally the user-cost of capital is determined for each subsector and each type of capital. The user-cost describes the cost of using one unit of capital for one period of time and is endogenous depending on investment prices.

The employment, *Q<k>*, *k= av,ak,as,ao,aq*, in each subsector is derived from the labour demand measured in hours per year, *Hq<k>*, as

(3.3.9) $Q<k>=\frac{Hq<k>}{Hgn}\cdot 1000$

where *Hgn* is the agreed number of working hours per year in the manufacturing industries in ADAM. This equation yields a rather rough estimate of the number of persons employed, because the amount of hours in manufacturing and agriculture are not necessarily the same.

The taxes paid and subsidies received are modelled in 4 groups. As mentioned above the modelling of taxes is by and large identical to the modelling of these variables concerning the aggregated agricultural sector in ADAM. Subsidies on production are considered exogenous, whereas the subsidies on products are modelled as a subsidy-rate times the production in fixed prices[1].

In general, output prices as well as factor prices except user-cost of capital are exogenous in the model. Current price projections are easily derived by inflating fixed price scenarios.

To enable the calculation of emissions from the agricultural sector the production in the subsectors are disaggregated to production measured in physical units at the ESMERALDA level, that is tons of crops in the *av*-subsector and number of animals in the *ak*-, *as*-, *ao*-subsectors. It is assumed that the tons produced and number of animals per volume of production is constant implying the physical production is proportional to production in fixed prices.

[1] The taxes modelled in LADA correspond with the indirect taxes in the ADAM input-output system. These are value added taxes, Siga<j>, taxes and subsidies on specific goods, Sipa<j>, and taxes and subsidies on production, Siqa<j>.

Finally, the LADA model contains some equations used to aggregate the subsector projections to a projection describing the agricultural sector in ADAM. This part of the model is referred to as *the aggregation module*. This module has two important properties

1) When using the LADA model for simulation in historical years on the constructed data, the results from the aggregation module concerning the entire agricultural sector in ADAM is in fact the historical observations of this sector
2) When using the model for aggregation of ESMERALDA scenarios, the subsector scenarios remain unaltered through simulation

The first property implies that the aggregation module and the data are consistant with the ADAM *a*-sector. The second property implies that it is in fact the LADA subsector scenarios based on the ESMERALDA scenarios that are aggregated even though, the LADA model has to simulate to derive the scenario describing the aggregated agricultural sector in ADAM

### *3.4 Using the LADA model*

Regardless of the use of the LADA model the output provided by the model is:

1) A scenario describing the activity in the agricultural sector in ADAM

2) Projections of the production in the ESMERALDA lines of production counted in produced tons in the crop lines and number of animals in the animal lines of production

The scenario describing the agricultural sector is used to analyse macroeconomic effects of some development in the agricultural sector. The productions in the ESMERALDA lines of production counted in physical units are used as input to the emission model calculating the emissions of $N_2O$, $CH_4$ etc. see Figure 1.1.
There are three different ways of using the LADA model:

1) Some ESMERALDA baseline scenario is aggregated to the ADAM level and used in an ADAM forecast

2) Given an ESMERALDA baseline scenario and one or more alternative scenarios macroeconomic and environmental effect of policies studied in ESMERALDA can be evaluated

3) Given an ESMERALDA baseline scenario the macroeconomic and environmental effects of restricting production in one or more of the subsectors: crops, cattle and milk, pigs and poultry can be studied using the LADA model only

Looking at case 1) the only task of the LADA model is to aggregate the five subsectors into a scenario describing ADAMs agricultural sector and pass series on production in physical terms to the emission model. Thereafter the environmental effects of the ESMERALDA forecast are calculated in the emission model, while some macroeconomic forecast based upon the ESMERALDA forecast of the agriculture can be made in ADAM.

In case 2) the objective will typically be to evaluate environmental benefits and economic costs from introducing some policy aimed at the agricultural sector. The ESMERALDA baseline scenario is used to construct consistent scenarios describing emissions and the macroeconomy. Then alternative emission and macroeconomic scenarios can be constructed consistent with the alternative ESMERALDA scenario. The environmental benefits can be assessed by comparing the baseline emission scenario to the alternative emission scenario. While economic costs in the agriculture can be evaluated by comparing the two ESMERALDA scenarios and derived macroeconomic effects can be found by comparing the macroeconomic baseline scenario to the alternative scenarios.

Case 3) is similar to case 2) except from the fact that only simple and small change in subsector production can be analysed using the LADA model alone, and that the economic influence from the change on the agriculture must be evaluated at the LADA or ADAM level.

**Annex 3.1**
**List of variables for LADA**

The notation is standard ADAM notation, so the only news for the ADAM-knowledgeable is the subsectors. A variable *X* appears normally in current prices, fixed prices, and as a deflator, notation is then *X, fX,* and *pX* respectively. The disaggregation of ADAM's a-sector implies that to the usual *a* for agriculture in ADAM will be added the following suffixes: *v, k, s, f,* and *q* for crops, cattle, pigs, poultry, and other agriculture respectively. Hence *fXas* is *X* in fixed prices for the pigs subsector.

io-coefficients have prefix *a* followed by supplying sector or import, and recipient sector, e.g. *anmas* the coefficient for supply from the *nm*-sector to subsector *s*.

**Variables**

*a<i><j>* *i=av, ak, as, af, aq, ng, ne, nf, nm, nt, nk, b, qh, qt, qq, m0, m2, m3k, m3q, m5, si, yw, yf*
*j=av, ak, as, af, aq*

coefficient for supply from sector *i* to use in sector *k*

Supplies are the same as standard ADAM-supply, except for the disaggregation of sector *a*.

| | |
|---|---|
| *av* | crops |
| *ak* | cattle |
| *as* | pigs |
| *af* | poultry |
| *aq* | others |
| *ng* | petroleum refineries |
| *ne* | energy suppliers |
| *nf* | manufacturing of food |
| *nm* | manufacturing of machinery |
| *nt* | transportation equipment |
| *nk* | chemical industry |
| *b* | construction |
| *qh* | trade |
| *qt* | other transport |
| *qq* | other services |
| *m0* | import of SITC 0: foodstuff |
| *m2* | import of SITC 2: unmanufactured goods, non food, except fuel |
| *m3k* | import of SITC 32: coal and coke |
| *m3q* | residual import of SITC 3: petroleum, electricity, and gas |
| *m5* | import of SITC 5: chemicals |
| *si* | indirect taxes, total |
| *yw* | compensation of employees |
| *yf* | gross value added |

*bhqa<j>* *j=v, k, s, f, q*

necessary input of hours per unit produced in sector *j*

*bivp<k>* *k=b, m*

present value of expected fiscal depreciation from an investment in capital type *k*

*bkba<j>* *j=v, k, s, f, q*
necessary input of buildings per unit produced in sector *j*

*bkma<j>* *j=v, k, s, f, q*
necessary input of equipment per unit produced in sector *j*

*bnva<j>* *j=v, k, s, f, q*
necessary input of land per unit produced in sector *j*

*bqsa*
ratio of self employed in ADAM's *a*-sector

*bqsa<j>* *j=v, k, s, f, q*
ratio of self employed in subsector *j*

*btgxa*
degree of charging VAT on ADAM's *a*-sector

*fIba<j>* *j=v, k, s, f, q*
gross fixed capital formation in buildings and civil engineering projects in subsector *j*, 1995 prices

*fIma<j>* *j=v, k, s, f, q*
gross fixed capital formation in machinery, transport equipment and other equipment in subsector *j*, 1995 prices

*fKba<j>* *j=v, k, s, f, q*
gross capital stock of buildings etc. in subsector *j*

*fKma<j>* *j=v, k, s, f, q*
gross capital stock of machinery etc. in subsector *j*, 1995 prices

*fKnba<j>* *j=v, k, s, f, q*
net capital stock of buildings etc. in subsector *j*

*fKnma<j>* *j=v, k, s, f, q*
net capital stock of machinery etc. in subsector *j*

*fVa<j>* *j=v, k, s, f, q*
use of energy and material in subsector *j*, 1995 prices

*fVea<j>* *j=v, k, s, f, q*
use of energy in subsector *j*, 1995 prices

*fVma<j>* *j=v, k, s, f, q*
use of materials in subsector *j*, 1995 prices

*fXa<j>* *j=v, k, s, f, q*

gross output in subsector *j,* 1995 prices

*fYfa*
gross value added in ADAM's *a*-sector, 1995 prices

*fYfa<j>* *j=v, k, s, f, q*
gross value added in subsector *j*, 1995 prices

*hgn*
average working hours in manufacturing, hours per year

*hqa<j>* *j=v, k, s, f, q*
volume of hours worked in subsector *j*

*Iba<j>j=v, k, s, f, q*
gross fixed capital formation in buildings and civil engineering projects in subsector *j*, current prices

*Ima<j>* *j=v, k, s, f, q*
gross fixed capital formation in machinery, transport equipment, and other equipment in subsector *j*, current prices

*iwbz*
redemption yields on bonds

*iwlo*
banks interest rate on advances

*la<j>* *j=v, k, s, f, q*
implicit hourly compensation per wage earner in subsector *j*

*n<j>* *j=km, ko, ka, kl, ss, sl, oe*
size of livestock in ESMERALDA subsector *j*

*n<j>e* *j=km, ko, ka, kl, ss, sl, oe*
size of livestock in ESMERALDA in subsector *j*, initial estimate for agruculture

*nv*
total land available

*nva<j>* *j=v, k*
land available to subsector *j*

*nvbr*
land lying fallow

*nv<j>* *j=sh, vg, rf*
hectares used in ESMERALDA subsector *j*

*nv<j>e* *j=sh, vg, rf*

hectares used in ESMERALDA subsector *j*, initial estimate for agriculture

*pwa<j>* *j=v, k, s, f, q*
average unit costs in subsector *j*

*pwaw*
average unit cost in ADAM's *a*-sector

*pIba*
price of buildings and civil engineering projects in ADAM's *a*-sector

*pIma*
price of machinery, transport equipment and other equipment in ADAM's *a*-sector

*pVa<j>* *j=v, k, s, f, q*
deflator for use of energy and materials in subsector *j*

*pVea<j>* *j=v, k, s, f, q*
deflator for use of energy in subsector *j*

*pVma<j>* *j=v, k, s, f, q*
deflator for use of materials in subsector *j*

*pXa<j>* *j=v, k, s, f, q*
deflator for gross output in subsector *j*

*pYfa<j>* *j=v, k, s, f, q*
deflator for gross value added in subsector *j*

*qsa<j>* *j=v, k, s, f, q*
number of self employed in subsector *j*

*qwa<j>* *j=v, k, s, f, q*
number of wage earners in subsector *j*

*rpi<k>ae* *k=b, m*
expected growth in *pi<k>a*

*Sigxa*
VAT revenue from gross output in ADAM's *a*-sector

*Sigxa<j>* *j=v, k, s, f, q*
VAT revenue from gross output in subsector *j*

*Sipvea*
revenue from duties on use of energy in ADAM's *a*-sector

*Sipvea<j>* *j=v, k, s, f, q*
net revenue from duties on use of energy in subsector *j*

*Sipxa*
net revenue from taxes on specific goods in ADAM's *a*-sector, total

*Sipxa<j>* *j=v, k, s, f, q*
net revenue from taxes on specific goods in subsector *j*, total

*Siqa*
net revenue from taxes on production in ADAM's *a*-sector, total

*Siqa<j>* *j=v, k, s, f, q*
net revenue from taxes on production in subsector *j*, total

*Siqal*
revenue from duties paid by employers on wage and salary costs in ADAM's *a*-sector

*Siqal<j>* *j=v, k, s, f, q*
revenue from duties paid by employers on wage and salary costs in subsector *j*

*tg*
VAT rate

*tsdsu*
expected marginal rate of corporation tax

*t<j>* *j=vf, vv, vh, vb, vo, vk, vr*
production in ESMERALDA subsector *j*, tons

*t<j>e* *j=vf, vv, vh, vb, vo, vk, vr*
production in ESMERALDA subsector *j*, tons, initial estimate for agriculture

*tvea*
rate of duty on *fVea*

*tvea<j>* *j=v, k, s, f, q*
rate of duty on *fVea<j>*

*ui<k>a* *k=b, m*
user-cost on capital stock of type *k,* in ADAM's *a*-sector

*ui<k>a<j>* *k=b, m* *j=v, k, s, f, q*
user-cost on capital stock of type *k,* in subsector *j*

*Va<j> j=v, k, s, f, q*
Use of energy and material in subsector *j*, current prices

*Vea<j>* *j=v, k, s, f, q*
Use of energy in subsector *j*, current prices

*Vma<j>* *j=v, k, s, f, q*
Use of materials in subsector *j*, current prices

*Xa<j>* *j=v, k, s, f, q*
gross output in subsector *j,* current prices

*Yfa*
gross value added in ADAM's *a*-sector, current prices

*Yfa<j>* *j=v, k, s, f, q*
gross value added in subsector *j*, current prices

*Ywa*
compensation of employees in ADAM's *a*-sector

*Ywa<j>* *j=v, k, s, f, q*
compensation of emplyees in subsector *j*

**() Annex 3.2**

() equations forming the LADA model

```
() ************
() SUB-SECTORS
() ************

() PRODUCTION

FRML _D    fXav = (nv-(nvsh+nvvg+nvrf)-nvbr)/bnvav $
FRML _D    fXak = nvak/bnvak                       $

FRML _I     Xav = pXav*fXav  $
FRML _I     Xak = pXak*fXak  $
FRML _I     Xas = pXas*fXas  $
FRML _I     Xao = pXao*fXao  $
FRML _I     Xaq = pXaq*fXaq  $


() ENERGY CONSUMPTION

FRML _GJR   fVeav  =  (angav+aneav+am3kav+am3qav)*fXav  $
FRML _GJR   fVeak  =  (angak+aneak+am3kak+am3qak)*fXak  $
FRML _GJR   fVeas  =  (angas+aneas+am3kas+am3qas)*fXas  $
FRML _GJR   fVeao  =  (angao+aneao+am3kao+am3qao)*fXao  $
FRML _GJR   fVeaq  =  (angaq+aneaq+am3kaq+am3qaq)*fXaq  $


FRML _I     Veav  =  pVeav*fVeav   $
FRML _I     Veak  =  pVeak*fVeak   $
FRML _I     Veas  =  pVeas*fVeas   $
FRML _I     Veao  =  pVeao*fVeao   $
FRML _I     Veaq  =  pVeaq*fVeaq   $


() MATERIAL COBSUMPTION

FRML _GJR   fVmav  =  (aavav+aakav+aasav+aaoav+aaqav+anfav+
                       anmav+antav+ankav+abav +aqhav+
                       aqtav+aqqav+am0av+am2av+am5av+
                       asiav)*fXav                           $

FRML _GJR   fVmak  =  (aavak+aakak+aasak+aaoak+aaqak+anfak+
                       anmak+antak+ankak+abak +aqhak+
                       aqtak+aqqak+am0ak+am2ak+am5ak+
                       asiak)*fXak                           $

FRML _GJR   fVmas  =  (aavas+aakas+aasas+aaoas+aaqas+anfas+
                       anmas+antas+ankas+abas +aqhas+
                       aqtas+aqqas+am0as+am2as+am5as+
                       asias)*fXas                           $

FRML _GJR   fVmao  =  (aavao+aakao+aasao+aaoao+aaqao+anfao+
                       anmao+antao+ankao+abao +aqhao+
                       aqtao+aqqao+am0ao+am2ao+am5ao+
                       asiao)*fXao                           $

FRML _GJR   fVmaq  =  (aavaq+aakaq+aasaq+aaoaq+aaqaq+anfaq+
                       anmaq+antaq+ankaq+abaq +aqhaq+
                       aqtaq+aqqaq+am0aq+am2aq+am5aq+
                       asiaq)*fXaq                           $
```

```
FRML _I     Vmav  =  pVmav*fVmav $
FRML _I     Vmak  =  pVmak*fVmak $
FRML _I     Vmas  =  pVmas*fVmas $
FRML _I     Vmao  =  pVmao*fVmao $
FRML _I     Vmaq  =  pVmaq*fVmaq $


() ENERGY- OG MATERIAL CONSUMPTION

FRML _I     fVav   =  fVmav + fVeav  $
FRML _I     fVak   =  fVmak + fVeak  $
FRML _I     fVas   =  fVmas + fVeas  $
FRML _I     fVao   =  fVmao + fVeao  $
FRML _I     fVaq   =  fVmaq + fVeaq  $

FRML _I     Vav    =  Vmav + Veav  $
FRML _I     Vak    =  Vmak + Veak  $
FRML _I     Vas    =  Vmas + Veas  $
FRML _I     Vao    =  Vmao + Veao  $
FRML _I     Vaq    =  Vmaq + Veaq  $

FRML _I     pVav    =  Vav/fVav  $
FRML _I     pVak    =  Vak/fVak  $
FRML _I     pVas    =  Vas/fVas  $
FRML _I     pVao    =  Vao/fVao  $
FRML _I     pVaq    =  Vaq/fVaq  $


() GROSS VALUE ADDED

FRML _I     Yfav   =  pYfav*fYfav  $
FRML _I     Yfak   =  pYfak*fYfak  $
FRML _I     Yfas   =  pYfas*fYfas  $
FRML _I     Yfao   =  pYfao*fYfao  $
FRML _I     Yfaq   =  pYfaq*fYfaq  $


() WAGES AND EMPLOYMENT

() bhqa<k> er det nødvendige timeinput pr. producerede enhed, er
endnu ikke dannet i banken
FRML _GJR  HQav = bhqav*fXav   $
FRML _GJR  HQak = bhqak*fXak   $
FRML _GJR  HQas = bhqas*fXas   $
FRML _GJR  HQao = bhqao*fXao   $
FRML _GJR  HQaq = bhqaq*fXaq   $

FRML _GJR  Qwav = HQav*(1-bqsav)*(1/hgn)*1000  $
FRML _GJR  Qwak = HQak*(1-bqsak)*(1/hgn)*1000  $
FRML _GJR  Qwas = HQas*(1-bqsas)*(1/hgn)*1000  $
FRML _GJR  Qwao = HQao*(1-bqsao)*(1/hgn)*1000  $
FRML _GJR  Qwaq = HQaq*(1-bqsaq)*(1/hgn)*1000  $

FRML _GJR  Qsav = HQav*bqsav*(1/hgn)*1000  $
FRML _GJR  Qsak = HQak*bqsak*(1/hgn)*1000  $
FRML _GJR  Qsas = HQas*bqsas*(1/hgn)*1000  $
FRML _GJR  Qsao = HQao*bqsao*(1/hgn)*1000  $
FRML _GJR  Qsaq = HQaq*bqsaq*(1/hgn)*1000  $
```

```
FRML _GJR  Ywav = lav*(1-bqsav)*hqav-Siqalv  $
FRML _GJR  Ywak = lak*(1-bqsak)*hqak-Siqalk  $
FRML _GJR  Ywas = las*(1-bqsas)*hqas-Siqals  $
FRML _GJR  Ywao = lao*(1-bqsao)*hqao-Siqalo  $
FRML _GJR  Ywaq = laq*(1-bqsaq)*hqaq-Siqalq  $


() CAPITAL, COSTS OF CAPITAL AND GROSS CAPITAL FORMATION

() GROSS CAPITAL STOCKS
FRML  _GJR  fKmav  =  fXav*bkmav $
FRML  _GJR  fKmak  =  fXak*bkmak $
FRML  _GJR  fKmas  =  fXas*bkmas $
FRML  _GJR  fKmao  =  fXao*bkmao $
FRML  _GJR  fKmaq  =  fXaq*bkmaq $

FRML  _GJR  fKbav  =  fXav*bkbav $
FRML  _GJR  fKbak  =  fXak*bkbak $
FRML  _GJR  fKbas  =  fXas*bkbas $
FRML  _GJR  fKbao  =  fXao*bkbao $
FRML  _GJR  fKbaq  =  fXaq*bkbaq $

() CAPITAL FORMATION
FRML  _I    fImav  =  fKmav-(1-bfimva)*fKmav(-1)  $
FRML  _I    fImak  =  fKmak-(1-bfimva)*fKmak(-1)  $
FRML  _I    fImas  =  fKmas-(1-bfimva)*fKmas(-1)  $
FRML  _I    fImao  =  fKmao-(1-bfimva)*fKmao(-1)  $
FRML  _I    fImaq  =  fKmaq-(1-bfimva)*fKmaq(-1)  $

FRML  _I    fIbav  =  fKbav-(1-bfibva)*fKbav(-1)  $
FRML  _I    fIbak  =  fKbak-(1-bfibva)*fKbak(-1)  $
FRML  _I    fIbas  =  fKbas-(1-bfibva)*fKbas(-1)  $
FRML  _I    fIbao  =  fKbao-(1-bfibva)*fKbao(-1)  $
FRML  _I    fIbaq  =  fKbaq-(1-bfibva)*fKbaq(-1)  $


FRML  _I    Imav  =  pImav*fImav  $
FRML  _I    Imak  =  pImak*fImak  $
FRML  _I    Imas  =  pImas*fImas  $
FRML  _I    Imao  =  pImao*fImao  $
FRML  _I    Imaq  =  pImaq*fImaq  $

FRML  _I    Ibav  =  pIbav*fIbav  $
FRML  _I    Ibak  =  pIbak*fIbak  $
FRML  _I    Ibas  =  pIbas*fIbas  $
FRML  _I    Ibao  =  pIbao*fIbao  $
FRML  _I    Ibaq  =  pIbaq*fIbaq  $


() NET CAPITAL STOCK
FRML _GJR   fKnmav =  fImav+(1-bfinmva)*fKnmav(-1) $
FRML _GJR   fKnmak =  fImak+(1-bfinmva)*fKnmak(-1) $
FRML _GJR   fKnmas =  fImas+(1-bfinmva)*fKnmas(-1) $
FRML _GJR   fKnmao =  fImao+(1-bfinmva)*fKnmao(-1) $
FRML _GJR   fKnmaq =  fImaq+(1-bfinmva)*fKnmaq(-1) $

FRML _GJR   fKnbav =  fIbav+(1-bfinbva)*fKnbav(-1) $
FRML _GJR   fKnbak =  fIbak+(1-bfinbva)*fKnbak(-1) $
FRML _GJR   fKnbas =  fIbas+(1-bfinbva)*fKnbas(-1) $
FRML _GJR   fKnbao =  fIbao+(1-bfinbva)*fKnbao(-1) $
FRML _GJR   fKnbaq =  fIbaq+(1-bfinbva)*fKnbaq(-1) $
```

```
FRML  _I    bfknmav = fKnmav/fKmav   $
FRML  _I    bfknmak = fKnmak/fKmak   $
FRML  _I    bfknmas = fKnmas/fKmas   $
FRML  _I    bfknmao = fKnmao/fKmao   $
FRML  _I    bfknmaq = fKnmaq/fKmaq   $

FRML  _I    bfknbav = fKnbav/fKbav   $
FRML  _I    bfknbak = fKnbak/fKbak   $
FRML  _I    bfknbas = fKnbas/fKbas   $
FRML  _I    bfknbao = fKnbao/fKbao   $
FRML  _I    bfknbaq = fKnbaq/fKbaq   $


() USER-COST
() MACHINERY
FRML  _GJR  uimav=bfknmav*pimav*(1-tsdsu*bivpm)
                        /(1-tsdsu)*((1-tsdsu)*iwlo+bfinmva-
0.5*rpimae)  $
FRML  _GJR  uimak=bfknmak*pimak*(1-tsdsu*bivpm)
                        /(1-tsdsu)*((1-tsdsu)*iwlo+bfinmva-
0.5*rpimae)  $
FRML  _GJR  uimas=bfknmas*pimas*(1-tsdsu*bivpm)
                        /(1-tsdsu)*((1-tsdsu)*iwlo+bfinmva-
0.5*rpimae)  $
FRML  _GJR  uimao=bfknmao*pimao*(1-tsdsu*bivpm)
                        /(1-tsdsu)*((1-tsdsu)*iwlo+bfinmva-
0.5*rpimae)  $
FRML  _GJR  uimaq=bfknmaq*pimaq*(1-tsdsu*bivpm)
                        /(1-tsdsu)*((1-tsdsu)*iwlo+bfinmva-
0.5*rpimae)  $

() BUILDINGS
FRML  _GJR  uibav=bfknbav*pibav*(1-tsdsu*bivpb)
                        /(1-tsdsu)*((1-tsdsu)*iwbz+bfinbva-
0.5*rpibae)  $
FRML  _GJR  uibak=bfknbak*pibak*(1-tsdsu*bivpb)
                        /(1-tsdsu)*((1-tsdsu)*iwbz+bfinbva-
0.5*rpibae)  $
FRML  _GJR  uibas=bfknbas*pibas*(1-tsdsu*bivpb)
                        /(1-tsdsu)*((1-tsdsu)*iwbz+bfinbva-
0.5*rpibae)  $
FRML  _GJR  uibao=bfknbao*pibao*(1-tsdsu*bivpb)
                        /(1-tsdsu)*((1-tsdsu)*iwbz+bfinbva-
0.5*rpibae)  $
FRML  _GJR  uibaq=bfknbaq*pibaq*(1-tsdsu*bivpb)
                        /(1-tsdsu)*((1-tsdsu)*iwbz+bfinbva-
0.5*rpibae)  $


() COSTS OF PRODUCTION

FRML  _GJR    pwav=(uimav*fKmav+uibav*fKbav+lav*Hqav+Veav+Vmav+siqav-
siqalv)/fXav  $
FRML  _GJR    pwak=(uimak*fKmak+uibak*fKbak+lak*Hqak+Veak+Vmak+siqak-
siqalk)/fXak  $
FRML  _GJR    pwas=(uimas*fKmas+uibas*fKbas+las*Hqas+Veas+Vmas+siqas-
siqals)/fXas  $
```

```
FRML  _GJR    pwao=(uimao*fKmao+uibao*fKbao+lao*Hqao+Veao+Vmao+siqao-
siqalo)/fXao  $
FRML  _GJR    pwaq=(uimaq*fKmaq+uibaq*fKbaq+laq*Hqaq+Veaq+Vmaq+siqaq-
siqalq)/fXaq  $


() TAXES AND DUTIES

FRML  _GJR  Sigxav = tg*btgxa*(1-tg*btgxa)*Vav $
FRML  _GJR  Sigxak = tg*btgxa*(1-tg*btgxa)*Vak $
FRML  _GJR  Sigxas = tg*btgxa*(1-tg*btgxa)*Vas $
FRML  _GJR  Sigxao = tg*btgxa*(1-tg*btgxa)*Vao $
FRML  _GJR  Sigxaq = tg*btgxa*(1-tg*btgxa)*Vaq $

FRML _GJR   Sipveav =  tveav*fVeav  $
FRML _GJR   Sipveak =  tveak*fVeak  $
FRML _GJR   Sipveas =  tveas*fVeas  $
FRML _GJR   Sipveao =  tveao*fVeao  $
FRML _GJR   Sipveaq =  tveaq*fVeaq  $

FRML _GJR   Sipxav = tvmav*fVmav + tveav*fVeav  $
FRML _GJR   Sipxak = tvmak*fVmak + tveak*fVeak  $
FRML _GJR   Sipxas = tvmas*fVmas + tveas*fVeas  $
FRML _GJR   Sipxao = tvmao*fVmao + tveao*fVeao  $
FRML _GJR   Sipxaq = tvmaq*fVmaq + tveaq*fVeaq  $

FRML _GJ_   Siqalv = Siqal*Qwav/(Qwav+Qwak+Qwas+Qwao+Qwaq)  $
FRML _GJ_   Siqalk = Siqal*Qwak/(Qwav+Qwak+Qwas+Qwao+Qwaq)  $
FRML _GJ_   Siqals = Siqal*Qwas/(Qwav+Qwak+Qwas+Qwao+Qwaq)  $
FRML _GJ_   Siqalo = Siqal*Qwao/(Qwav+Qwak+Qwas+Qwao+Qwaq)  $
FRML _GJ_   Siqalq = Siqal*Qwaq/(Qwav+Qwak+Qwas+Qwao+Qwaq)  $



() ********************************
() AGGREGATION TO THE AGRICULTURAL SECTOR IN ADAM
() ********************************

() PRODUCTION

FRML _I  fXa = fXav + fXak + fXas + fXao + fXaq  $
FRML _I  Xa  =  Xav +  Xak +  Xas +  Xao + Xaq  $
FRML _I  pXa =  Xa/fXa $


() ENERGY- og MATERIAL CONSUMPTION

FRML _I  fVea  =  fVeav + fVeak + fVeas + fVeao + fVeaq  $
FRML _I   Vea  =   Veav +  Veak +  Veas +  Veao + Veaq  $
FRML _I  pVea  =  Vea/fVea $

FRML _I  fVma  =  fVmav + fVmak + fVmas + fVmao + fVmaq  $
FRML _I   Vma  =   Vmav +  Vmak +  Vmas +  Vmao +  Vmaq  $
FRML _I  pVma  =  Vma/fVma $

FRML _I  fVa   =  fVav + fVak + fVas + fVao + fVaq  $
FRML _I   Va   =   Vav +  Vak +  Vas +  Vao + Vaq  $
FRML _I  pVa   =   Va/fVa $
```

() GROSS VALUE ADDED

```
FRML _I  fYfa  =  fYfav + fYfak + fYfas + fYfao + fYfaq  $
FRML _I   Yfa  =   Yfav +  Yfak +  Yfas +  Yfao + Yfaq  $
FRML _I  pYfa  =  Yfa/fYfa  $
```

() WAGE AND EMPLOYMENT

```
FRML _I  HQa   =  HQav + HQak + HQas + HQao+ HQaq  $
FRML _I  Qwa   =  Qwav + Qwak + Qwas + Qwao+ Qwaq  $
FRML _I  Qsa   =  Qsav + Qsak + Qsas + Qsao+ Qsaq  $
FRML _I  Ywa   =  Ywav + Ywak + Ywas + Ywao+ Ywaq  $
```

() CAPITAL, CAPITAL COSTS AND GROSS CAPITAL FORMATION

```
FRML _I  fKma  =  fKmav + fKmak + fKmas + fKmao + fKmaq  $
FRML _I  fKba  =  fKbav + fKbak + fKbas + fKbao + fKbaq  $

FRML _I  fKnma  =  fKnmav + fKnmak + fKnmas + fKnmao + fKnmaq  $
FRML _I  fKnba  =  fKnbav + fKnbak + fKnbas + fKnbao + fKnbaq  $

FRML _I  fIma  =  fImav + fImak + fImas + fImao + fImaq  $
FRML _I  fIba  =  fIbav + fIbak + fIbas + fIbao + fIbaq  $
FRML _I   Ima  =   Imav +  Imak +  Imas +  Imao +  Imaq  $
FRML _I   Iba  =   Ibav +  Ibak +  Ibas +  Ibao +  Ibaq  $
FRML _I  pIma  =  Ima/fIma  $
FRML _I  pIba  =  Iba/fIba  $

FRML _I  bfknma = fKnma/fKma $
FRML _I  bfknba = fKnba/fKba $

FRML _I  uima   = bfknma*pima*(1-tsdsu*bivpm)/(1-tsdsu)
                        *((1-tsdsu)*iwlo+bfinmva-0.5*rpimae)  $
FRML _I  uiba   = bfknba*piba*(1-tsdsu*bivpb)/(1-tsdsu)
                        *((1-tsdsu)*iwbz+bfinbva-0.5*rpibae)  $

FRML _I  la     =(lav*Hqav+lak*Hqak+las*Hqas+lao*Hqao+laq*Hqaq)/hqa $
```

() COSTS OF PRODUCTION

```
FRML _GJR pwaw= (uima*fKma+uiba*fKba+la*hqa+Vea+Vma+Siqa-siqal)/fXa $
```

() TAXES AND DUTIES

```
FRML _I  Sigxa  =  Sigxav + Sigxak + Sigxas + Sigxao + Sigxaq  $
FRML _I  Sipxa  =  Sipxav + Sipxak + Sipxas + Sipxao + Sipxaq  $
FRML _I  Sipvea =  Sipveav+ Sipveak+ Sipveas+ Sipveao+ Sipveaq $

FRML _I  tvma   =  (Sipxa-Sipvea)/fVma $
FRML _I  tvea   =  Sipvea/fVea $

FRML _I  Siqal  =  Siqalv + Siqalk + Siqals + Siqalo + Siqalq  $
FRML _I  Siqa   =  Siqav  + Siqak  + Siqas  + Siqao  + Siqaq   $
```

```
() ***********************************************
() DISAGGREGATION TO PHYSICAL ESMERALDA VARIABLES
() ***********************************************


FRML _GJR  nkm  =  nkme*(fXak/fXake)          $
FRML _GJR  nko  =  nkoe*(fXak/fXake)          $
FRML _GJR  nka  =  nkae*(fXak/fXake)          $
FRML _GJR  nkl  =  nkle*(fXak/fXake)          $
FRML _GJR  nss  =  nsse*(fXas/fXase)/1.9      $
FRML _GJR  nsl  =  nsle*(fXas/fXase)/1.9      $
FRML _GJR  noe  =  noee*(fXas/fXase)*6.421   $
FRML _GJR  tvf  =  tvfe*(fXav/fXave)          $
FRML _GJR  tvv  =  tvve*(fXav/fXave)          $
FRML _GJR  tvh  =  tvhe*(fXav/fXave)          $
FRML _GJR  tvb  =  tvbe*(fXav/fXave)          $
FRML _GJR  tvo  =  tvoe*(fXav/fXave)          $
() FRML _GJR  tvg  =  tvg*(fXav/fXave)          $
FRML _GJR  tvk  =  tvke*(fXav/fXave)          $
FRML _GJR  tvr  =  tvre*(fXav/fXave)          $
() FRML _GJR  tvrf = tvrf*(fXav/fXave)          $
() FRML _GJR  tvsh = tvsh*(fXav/fXave)          $
() FRML _GJR  tvvg = tvvg*(fXav/fXave)          $
() FRML _GJR  tvoc = tvoc*(fXav/fXave)          $
() FRML _GJR  nvg  = nvge*fXav/fXave $
FRML _GJR  nvsh = nvshe*nvak/nvake  $
FRML _GJR  nvvg = nvvge*nvak/nvake  $
FRML _GJR  nvrf = nvrfe*nvak/nvake  $
```